Paper 209-25

An Interactive Macro Program for Line × Tester Analysis

Violeta Bartolome, Biometrics, International Rice Research Institute (IRRI), Los Baños, Philippines
Glenn Gregorio, Plant Breeding, Genetics and Biochemistry Division, IRRI, Los Baños, Philippines

ABSTRACT

Breeders use line × tester analysis to evaluate the general and specific combining ability of various lines and to estimate gene effects. This analysis requires a lengthy series of complicated computational steps, which when done in SAS® would require in-depth knowledge of the software. However, very few breeders have this capability; thus a macro program was developed for users who have limited knowledge of SAS.

This program uses the Macro and Data Step Report Writing facilities in Base SAS to request information interactively and print out the needed analyses. The required statistical analyses were generated using the SAS/STAT® module. You only need to know how to open the program file, invoke the macro, respond to queries, and print the output. This paper is intended for all levels of SAS users.

INTRODUCTION

The Biometrics Unit is one of the research service units of the International Rice Research Institute. It assists researchers in planning experiments, collecting and analyzing data, and interpreting and presenting results. The Plant Breeding, Genetics, and Biochemistry Division is one key research division that is supported by Biometrics. In this division, researchers and breeders require tools to help them in the analysis of their research data. One of these tools is the line × tester analysis.

The line × tester analysis involves a lengthy series of complicated computational steps, which when done in SAS would require in-depth knowledge of the software. Thus to help breeders analyze their own data, we developed interactive programs such as that for line × tester analysis using the SAS macro facility. The SAS/AF® is not available at the Institute. The SAS macro facility is the only available tool that could be used to develop application programs.

In formulating programs for genetic improvement of yield, a breeder faces the difficult task of choosing parents for hybridization. This is because yield is a complex character, comprising several components, each of which is polygenically controlled and, therefore, very susceptible to environmental fluctuations (Singh 1966). The task becomes even more complex when the breeder has to choose from a wide collection of diverse germplasm in his breeding station. In predominantly self-pollinated crops, the technique of line × tester analysis has proved useful for screening lines rapidly and with a reasonable degree of confidence. Breeders use the line × tester analysis to evaluate the general and specific combining ability of parents. At the same time, this analytical tool is useful in estimating various types of gene effects. The analysis provides an opportunity for discriminating large numbers of parents for their combining ability without making so many crosses. Prasad and Sastry (1987) and Manuel and Palanisamy (1989) used this technique to identify parents and crosses that could be exploited for future breeding programs. The tables presented in these papers are uniform and thus can be standardized in a macro program.

THE SAS PROGRAM FOR LINE × TESTER ANALYSIS

The step-by-step procedure presented by Singh and Chaudhary (1977) was used in developing the program for line × tester analysis, with some modifications in the partitioning of variety effect. The program was developed using the macro facility, a powerful tool in SAS programming. To illustrate the program, we used a sample data set with 7 lines, 4 testers, and 4 checks. The program code is in Appendix A and sample outputs are presented in Appendix B.

The program was developed under the Windows environment. It is interactive and does not require proficiency in the use of SAS. You only needs to know how to open the program file, invoke the macro, respond to queries, and print the output. However, if you intend to write a similar program, you will need an in-depth knowledge of the macro facility.

The program, through messages flashed in the log window, requests you to input information in the program editor and then to submit responses by clicking on the running man icon. It uses the %PUT to display instructions in the log window and the %INPUT to assign the values you have inputted to macro variables. The information required are number of testers, lines, checks, and replicates and the start and end column of their location in the input file. The program also asks for the name of the data set to be analyzed; entry number of lines, testers, and checks; and the number and names of characters to be analyzed. In response to queries on entry numbers, you should enter the numbers as a string separated by a period (e.g., 5.6.7.8.9.10.11). The program uses the %LET, %INDEX, %EVAL, and %SUBSTR to decipher the different entry numbers. These information are needed to create subset data sets for lines, testers, and checks that will be needed in the computation of various estimates. The %DO-%END is used to combine variable numbers of data sets.

OUTPUTS

After the program has received the needed information, it outputs tables needed in a line × tester analysis. The outputs of the program include the following:

1. Analysis of variance with partitioned variety group effects. This helps in sorting out the variance due to different sources. The variations were partitioned between varieties and within variety groups. This provides a basis for the test of significance. The general linear models (GLM) procedure of SAS/STAT was used to output the sums of squares needed, including those for the partitioned effects. The program was developed to handle any number of testers, lines, and checks. The contrast statement of the PROC GLM was not used to partition variety effect as the number of contrast coefficients will depend on the number of testers, lines, and checks. All sums of squares were combined in one data set. F values and their corresponding levels of significance were recomputed and then printed in a table using the data step report writing facility.
2. Table of parent means and their respective general combining ability (GCA), and hybrid means and their respective specific combining ability (SCA). GCA is the average performance of a line in a series of hybrids and represents additive gene action (fixable genes). SCA measures the deviation of hybrids from the value expected on the basis of parental performance. It represents nonadditive gene action (nonfixable genes). Combining ability studies help in identifying parents with high GCA and in identifying cross combinations showing high SCA. The GCA is the difference of the parent mean from the grand mean, while the SCA is the hybrid mean minus the line and tester effects. The MEANS procedure was used to output the needed means which were then merged into one data set. Arithmetic operations were performed to derive the needed combining ability effects, while the TABULATE procedure was used to generate the tables.
3. Table of standard errors for combining ability effect. These standard errors can be used to test the significance of the different combining ability effects. Arithmetic operations were

made to compute for the standard errors needed to compare GCAs and SCAs. The data step report writing facility was used to generate the table.

4. Heterosis analysis. This table presents the performance of the hybrids compared with their parents. It can be used to identify hybrid performance that exceeds the average parental performance. Arithmetic operations were performed to come up with different parameters and the TABULATE procedure was used to generate the table.

CONCLUSION

The use of the SAS software need not be limited to users who are proficient in SAS programming. Programs may be developed such that others who have limited knowledge of SAS may also use the software in data analysis without asking help of a statistician. If the SAS/AF is not available, the SAS macro facility could also be used to develop applications programs.

TRADEMARKS

REFERENCES

Manuel WW, Palanisamy S. 1989. Line x tester analysis of combining ability of rice. Oryza. 26: 27-32.

Prasad GSV, Sastry MVS. 1987. Line x tester analysis for combining ability and heterosis in brown planthopper-resistant varieties. Indian Agric. 31: 257-265.

Singh RK, Chaudhary BD. 1979. Biometrical methods in quantitative genetic analysis. New Delhi: Kalyani Publishers.

Singh SP, Joshi AB. 1966. Line × tester analysis in relation to breeding for yield in linseed. Indian J. Genet. Plant Breed. 26: 177-194.

APPENDIX A. PROGRAM CODE

```
%MACRO LINXTES;

*---- MACROS USED ----*;

%MACRO ULIT1(TRT);
     CLASS REP &TRT;
     MODEL Y=REP &TRT REP*&TRT/SS3;
%MEND ULIT1;
RUN;
%MACRO ULIT2(SOURCE,S);
DATA &SOURCE; SET LINE;
     IF _SOURCE_="&S";     MS=SS/DF;
DATA &SOURCE; SET &SOURCE; LENGTH SOURCE $20;
     SOURCE="&SOURCE";
%MEND ULIT2;
RUN;
%MACRO ULIT3(DATA1,FROM,DATA2,SOURCE,SOURCE2);
DATA &DATA1; SET &FROM;
     IF _SOURCE_="&SOURCE";     MS=SS/DF;
DATA &DATA1; SET &DATA1;
DATA &DATA1; MERGE &DATA1 &DATA2;
     LENGTH SOURCE $ 20;     SOURCE=&SOURCE2;
     F=MS/MSE;
     A1=2/9/DF; B1=2/9/EDF;
     Z=(ABS((1-B1)*(F**(1/3))-1+A1))/
       (SQRT(B1*(F**(2/3))+A1));
     IF EDF<4 THEN
        Z=Z*((1+(.08*(Z**4)))/(EDF**3));
     PROB=0.5/    ((1+(Z*(.196854+(Z*(.115194
        +(Z*(.000344+(Z*.019527))))))))**4);
     KEEP SOURCE DF SS MS F PROB;
%MEND ULIT3;
RUN;
%PUT ****INPUT NUMBER OF LINES*****;
%INPUT L;
%PUT ****INPUT NUMBER OF TESTERS*****;
%INPUT T;
%PUT ****INPUT NUMBER OF CHECKS*****;
%INPUT C;
%PUT ****INPUT NUMBER OF REPS*****;
%INPUT REP;
%PUT
****INPUT RAW DATA FILE REF (EX: A:RAW.DAT)****;
%INPUT FILEREF;
%PUT
**INPUT START AND END COL FOR VARIETY ID
  (EX: 3 5)**;
%INPUT STRT ETRT;
%PUT ***INPUT START AND END COLUMNS FOR REP***;
%INPUT SREP EREP;
%PUT **INPUT NUMBER OF VARIABLES TO BE ANALYZED
     (Y)**;
%INPUT N;
%DO J=1 %TO &N;
   %PUT **INPUT START AND END COLUMNS FOR Y&J**;
   %INPUT SX EX;
   %LET S&J=%EVAL(&SX+0);
   %LET E&J=%EVAL(&EX+0);
   %PUT ****INPUT NAME OF Y&J ****;
   %INPUT;
   %LET CHAR&J=&SYSBUFFR;
%END;

OPTIONS CHARCODE MISSING=' ' REPLACE NONUMBER
NODATE;
TITLE;

%LET N1=%EVAL(&N-1);

PROC FORMAT;
     VALUE CHARFMT
          %DO I=1 %TO &N;
              &I="&&CHAR&I"
           %END;
           ;
     VALUE IDFMT 1='TESTER'        2='LINE';

DATA LINXTES; INFILE "&FILEREF";
     INPUT VAR &STRT-&ETRT REP &SREP-&EREP @;
     %DO I=1 %TO &N1;
     INPUT X&I &&S&I-&&E&I @;
     %END;
     INPUT X&N &&S&N-&&E&N;
     %DO I=1 %TO &N;
         LABEL X&I="&&CHAR&I";
     %END;

***** CREATION OF PARENT (LINES) DATASET *****;

%PUT  **INPUT ID NOS. OF PARENT VARIETIES WHICH
      ARE LINES**;
%PUT  ***SEPARATE ENTRIES WITH PERIODS***;
%INPUT TEMP;
%DO I=1 %TO &L-1;
   %LET POS=%INDEX(&TEMP,.);
   %LET S=%EVAL(&POS-1);
   %LET L&I=%SUBSTR(&TEMP,1,&S);
   %LET S=%EVAL(&POS+1);
   %LET TEMP=%SUBSTR(&TEMP,&S);
%END;
  %LET L&L=&TEMP;
%DO I=1 %TO &L;
   DATA LINE&I; SET LINXTES; IF VAR=&&L&I;
   %IF &I=2 %THEN %DO;
       DATA LINE; SET LINE1 LINE2;
   %END;
   %IF &I>2 %THEN %DO;
       DATA LINE; SET LINE LINE&I;
   %END;
%END;
DATA LINE; SET LINE;
     ID=2;
```

```
***** CREATION OF PARENT (TESTER) DATASET *****;

%PUT  ***INPUT ID NOS. OF PARENT VARIETIES WHICH
      ARE TESTERS***;
%PUT  ***SEPARATE ENTRIES WITH PERIODS***;

%INPUT TEMP;
%DO I=1 %TO &T-1;
   %LET POS=%INDEX(&TEMP,.);
   %LET S=%EVAL(&POS-1);
   %LET T&I=%SUBSTR(&TEMP,1,&S);
   %LET S=%EVAL(&POS+1);
   %LET TEMP=%SUBSTR(&TEMP,&S);
%END;
  %LET T&T=&TEMP;
%DO I=1 %TO &T;
   DATA TESTER&I; SET LINXTES; IF VAR=&&T&I;
   %IF &I=2 %THEN %DO;
       DATA TESTER; SET TESTER1 TESTER2;
   %END;
   %IF &I>2 %THEN %DO;
       DATA TESTER; SET TESTER TESTER&I;
   %END;
%END;
DATA TESTER; SET TESTER;
     ID=1;
DATA PARENT; SET TESTER LINE;
     S='P';

******** CREATION OF TESTER DATASET *********;

%DO M=1 %TO &T;
%PUT  ***INPUT ID NOS. OF HYBRID VARIETIES WITH
      TESTER&M AS PARENT***;
%PUT  ***SEPARATE ENTRIES WITH PERIODS***;

%INPUT TEMP;
%DO I=1 %TO &L-1;
   %LET POS=%INDEX(&TEMP,.);
   %LET S=%EVAL(&POS-1);
   %LET V&I=%SUBSTR(&TEMP,1,&S);
   %LET S=%EVAL(&POS+1);
   %LET TEMP=%SUBSTR(&TEMP,&S);
%END;
  %LET V&L=&TEMP;
%DO I=1 %TO &L;
   DATA T&I; SET LINXTES; IF VAR=&&V&I;
   %IF &I=2 %THEN %DO;
       DATA T; SET T1 T2;
   %END;
   %IF &I>2 %THEN %DO;
       DATA T; SET T T&I;
   %END;
%END;
DATA TESTER&M; SET T;
     TESTER=&&T&M;
%IF &M=2 %THEN %DO;
    DATA TESTER; SET TESTER1 TESTER2;
%END;
%IF &M>2 %THEN %DO;
    DATA TESTER; SET TESTER TESTER&M;
%END;
%END;
PROC SORT; BY VAR REP;

********** CREATION OF LINE DATASET ***********;

%DO M=1 %TO &L;
%PUT  ***INPUT ID NOS. OF HYBRID VARIETIES WITH
      LINE&M AS PARENT***;
%PUT  ***SEPARATE ENTRIES WITH PERIODS***;

%INPUT TEMP;
%DO I=1 %TO &T-1;
   %LET POS=%INDEX(&TEMP,.);
   %LET S=%EVAL(&POS-1);
   %LET V&I=%SUBSTR(&TEMP,1,&S);
   %LET S=%EVAL(&POS+1);
   %LET TEMP=%SUBSTR(&TEMP,&S);
%END;
  %LET V&T=&TEMP;
%DO I=1 %TO &T;
   DATA L&I; SET LINXTES; IF VAR=&&V&I;
   %IF &I=2 %THEN %DO;
       DATA L; SET L1 L2;
   %END;
   %IF &I>2 %THEN %DO;
       DATA L; SET L L&I;
   %END;
%END;
DATA LINE&M; SET L;
     LINE=&&L&M;
%IF &M=2 %THEN %DO;
    DATA LINE; SET LINE1 LINE2;
%END;
%IF &M>2 %THEN %DO;
    DATA LINE; SET LINE LINE&M;
%END;
%END;
PROC SORT DATA=LINE; BY VAR REP;

******** CREATION OF CHECK DATASET *********;

%PUT  ***INPUT ID NOS. OF CHECK VARIETIES***;
%PUT  ***SEPARATE ENTRIES WITH PERIODS***;

%INPUT TEMP;
%DO I=1 %TO &C-1;
   %LET POS=%INDEX(&TEMP,.);
   %LET S=%EVAL(&POS-1);
   %LET C&I=%SUBSTR(&TEMP,1,&S);
   %LET S=%EVAL(&POS+1);
   %LET TEMP=%SUBSTR(&TEMP,&S);
%END;
  %LET C&C=&TEMP;
%DO I=1 %TO &C;
   DATA CHECK&I; SET LINXTES; IF VAR=&&C&I;
   %IF &I=2 %THEN %DO;
       DATA CHECK; SET CHECK1 CHECK2;
   %END;
   %IF &I>2 %THEN %DO;
       DATA CHECK; SET CHECK CHECK&I;
   %END;
%END;
DATA CHECK; SET CHECK;
     S='C';
DATA PC; SET PARENT CHECK;
PROC SORT; BY VAR REP;
DATA HYBRID; MERGE LINE TESTER; BY VAR REP;
     S='H';
DATA LINXTES; MERGE PC HYBRID; BY VAR REP;

%DO K=1 %TO &N;

DATA TEMP; SET LINXTES;
     RENAME X&K=Y;       LABEL X&K="&&CHAR&K";

********** CREATION OF MAIN ANOV ************;

PROC GLM NOPRINT DATA=TEMP OUTSTAT=LINE;
     %ULIT1(VAR);

***** SS OF CHECKS VS PARENTS VS HYBRIDS *****;

PROC GLM NOPRINT DATA=TEMP OUTSTAT=CPH;
     %ULIT1(S);

***** SS OF CHECKS VS (PARENTS & HYBRIDS) *****;

DATA C2; SET TEMP;
     IF S='P' OR S='H' THEN S='H';
PROC GLM NOPRINT OUTSTAT=CVPH;
     %ULIT1(S);

********** SS OF PARENTS VS HYBRIDS **********;

DATA C3; SET TEMP;
```

```
     IF S='P' OR S='H';
PROC GLM NOPRINT OUTSTAT=PH;
     %ULIT1(S);

********** SS OF CHECKS *************;

DATA CHECK; SET TEMP;
     IF S='C';
PROC GLM NOPRINT OUTSTAT=C;
     %ULIT1(VAR);

********** SS OF PARENTS ************;

DATA PARENT; SET TEMP;
     IF S='P';
PROC GLM NOPRINT DATA=PARENT OUTSTAT=P;
     %ULIT1(VAR);

********** SS OF HYBRIDS *************;

DATA HYBRID; SET TEMP;
     IF S='H';
PROC GLM NOPRINT DATA=HYBRID OUTSTAT=H;
     %ULIT1(VAR);

******** SS OF LINE, TESTER AND L X T *******;

PROC GLM NOPRINT DATA=HYBRID OUTSTAT=LXT;
     CLASS REP LINE TESTER;
     MODEL Y=REP LINE ?/ TESTER/SS3;

********** ANOVA TABLE ************;

%ULIT2(ERROR,REP*VAR);
DATA E; SET ERROR;
     RENAME MS=MSE DF=EDF;          KEEP DF MS;

%ULIT3(CPH,CPH,E,S," C VS P VS H");
%ULIT3(REP,LINE,E,REP,"REP");
%ULIT3(VAR,LINE,E,VAR,"VAR");
%ULIT3(CVPH,CVPH,E,S,"  C VS (P,H)");
%ULIT3(PH,PH,E,S,"  P VS H");
%ULIT3(C,C,E,VAR," CHECKS (C)");
%ULIT3(P,P,E,VAR," PARENT (P)");
%ULIT3(H,H,E,VAR," HYBRID (H)");
%ULIT3(LXT2,LXT,E,LINE*TESTER,"  LINE x
TESTER");

DATA E2; SET LXT2;
     RENAME MS=MSE DF=EDF;      KEEP DF MS;

%ULIT3(LINE,LXT,E2,LINE,"  LINE");
%ULIT3(TESTER,LXT,E2,TESTER,"  TESTER");

PROC MEANS MEAN NOPRINT DATA=TEMP;
     VAR Y;     OUTPUT OUT=GMEAN MEAN=GMEAN;
DATA CV; IF _N_=1 THEN SET GMEAN; SET E;
     CV=(SQRT(MSE)/GMEAN)*100;     KEEP CV;

DATA ANOV; SET REP VAR CPH CVPH PH C P H LINE
TESTER LXT2 ERROR CV;

TITLE;

DATA _NULL_; SET ANOV END=EOF;
R1=REPEAT('-',78);
  FILE PRINT NOTITLES HEADER=H;
  PUT @ 5 SOURCE $CHAR20. @;
     IF SOURCE =: ' ' AND SOURCE NE: '  ' THEN
DO;
        PUT @ 29 DF 4. @ 37 SS 13.4 @ 52 MS 13.4
@; END;
     IF SOURCE =: '  ' THEN DO;
        PUT @ 30 DF 4. @ 37 SS 13.4 @ 52 MS 13.4
@; END;
     IF SOURCE NE: ' ' AND SOURCE NE: '  ' THEN
DO;
        PUT @ 28 DF 4. @ 37 SS 13.4 @ 52 MS 13.4
@; END;
     IF F < 1 AND F NE . THEN PUT @ 71 '< 1';
     ELSE DO; PUT @ 66 F 9.2 @ 78 PROB 6.4; END;
     IF EOF THEN PUT @5 R1 /
         @5 'CV = '@ 10 CV 5.2 @ 15 '%';
    RETURN;
H:  PUT /////
    @23 "ANALYSIS OF VARIANCE FOR &&CHAR&K" //
    @5 R1 /
    @5 'Source'              @42 'Sum' /
    @5 'of'                  @43 'of'      @57
'Mean' /
    @5 'Variation' @30 'df'  @40 'Squares' @56
'Square' @72 'F'  @78 'Prob'/
    @5 R1 ; RETURN;

***** ESTIMATION OF GCA EFFECTS FOR LINES *****;

PROC MEANS MEAN NOPRINT DATA=HYBRID;
     VAR Y; OUTPUT OUT=GMEAN MEAN=GMEAN;

PROC SORT DATA=HYBRID; BY LINE;
PROC MEANS MEAN NOPRINT DATA=HYBRID; BY LINE;
     VAR Y; OUTPUT OUT=LMEAN MEAN=LMEAN;

DATA GCA_L; IF _N_=1 THEN SET GMEAN; SET LMEAN;
     GCA=LMEAN-GMEAN;     VAR=LINE;

***** ESTIMATION OF GCA EFFECTS FOR LINE *****;

PROC SORT DATA=HYBRID; BY TESTER;
PROC MEANS MEAN NOPRINT DATA=HYBRID; BY TESTER;
     VAR Y; OUTPUT OUT=TMEAN MEAN=TMEAN;

DATA GCA_T; IF _N_=1 THEN SET GMEAN; SET TMEAN;
     GCA=TMEAN-GMEAN;     VAR=TESTER;

DATA GCA; SET GCA_L GCA_T;
PROC SORT; BY VAR;
PROC SORT DATA=PARENT; BY VAR;
PROC MEANS MEAN NOPRINT DATA=PARENT; BY VAR;
     ID ID;
     VAR Y; OUTPUT OUT=PMEAN MEAN=PMEAN;

DATA GCA; MERGE PMEAN GCA; BY VAR;
     FORMAT ID IDFMT.;

****** TABULATION OF PARENTS MEANS & GCA ******;

PROC TABULATE ;
     VAR PMEAN GCA;
     CLASS VAR ID;
     TITLE1 "TABLE OF PARENT MEANS AND GCA FOR
&&CHAR&K";
     TITLE2 "(AVE. OF &REP REPS)";
     TABLE ID=' '*VAR=' ',(PMEAN='Parent
           Mean'*F=10.3 GCA*F=12.4)*MEAN=' '/
           BOX="                       Variety";
RUN; TITLE;

******** FOR ESTIMATION OF SCA EFFECTS *******;

PROC SORT DATA=HYBRID; BY LINE TESTER;
PROC MEANS MEAN NOPRINT DATA=HYBRID;
     BY LINE TESTER;
     VAR Y; OUTPUT OUT=HMEAN MEAN=HMEAN; ID VAR;
DATA SCA; MERGE HMEAN LMEAN; BY LINE;
PROC SORT; BY TESTER;
DATA SCA; MERGE SCA TMEAN; BY TESTER;
DATA SCA; IF _N_=1 THEN SET GMEAN; SET SCA;
     SCA=HMEAN-LMEAN-TMEAN+GMEAN;

*** TABULATION OF HYBRID MEANS & TREATMENT ***;

PROC TABULATE ;
     VAR HMEAN SCA;
     CLASS VAR;
     TITLE1 "TABLE OF HYBRID MEANS AND SCA FOR
             &&CHAR&K";
     TITLE2 "(AVE. OF &REP REPS)";
```

```
      TABLE VAR=' ',(HMEAN='Hybrid Mean'*F=10.3
             SCA*F=11.4)*MEAN=' '/BOX="Variety";
      RUN; TITLE;

** COMPUTATION OF THE STANDARD ERRORS FOR
   COMBINING ABILITY EFFECTS **;
DATA L; SET LINE;   RENAME MS=ML SS=SSL;
DATA H2; SET H;     RENAME MS=MH SS=SSH;
DATA T; SET TESTER; RENAME MS=MT SS=SST;
DATA LXT; SET LXT2; RENAME MS=MLXT SS=SSLXT;
DATA E; SET ERROR;  RENAME MS=ME SS=SSE;
DATA SE; MERGE L H2 T LXT E;
      SE1=SQRT(ME/(&REP*&T));
SE2=SQRT(ME/(&REP*&L));
      SE3=SQRT(ME/&REP);
SE4=SQRT(2*ME/(&REP*&T));
      SE5=SQRT(2*ME/(&L*&REP));
SE6=SQRT(2*ME/&REP);
      COV1=(ML-MLXT)/(&REP*&T);
      COV2=(MT-MLXT)/(&REP*&L);
      COV3=(1/(&REP*(2*&L*&T-&L-&T)))*
          (((((&L-1)*ML)+((&T-1)*MT))/(&L+&T-2))-
           MLXT);
      COV4=(((ML-ME)+(MT-ME)+(MLXT-ME))/
          (3*&REP))+(((6*&REP*COV3)-(&REP*
          (&L+&T)*COV3))/(3*&REP));
      S2A0=16*COV3;
      S2A1=(16*COV3)/4;
      S2SCA=(MLXT-ME)/&REP;
      S2D0=16*S2SCA;
      S2D1=(16*S2SCA)/4;
      CL=(SSL*100)/SSH;
      CT=(SST*100)/SSH;
      CLXT=(SSLXT*100)/SSH;

DATA _NULL_; SET SE;
R1=REPEAT('-',46);
FILE PRINT;
PUT ////
@25 'STANDARD ERRORS OF COMBINING ABILITY
EFFECTS' /
@33 "FOR &&CHAR&K"  ///
@24 R1 /
@25 'Parameter'
@62 'Estimate' /   @24 R1 /
@25 'S.E. (gca for line)'
@55 SE1     15.4 /
@25 'S.E. (gca for tester)'
@55 SE2     15.4 /
@25 'S.E. (sca for effects)'
@55 SE3     15.4 /
@25 'S.E. (gi-gj) line'
@55 SE4     15.4 /
@25 'S.E. (gi-gj) tester'
@55 SE5     15.4 /
@25 'S.E. (sij-skl)'
@55 SE6     15.4 /
@25 'Cov. H.S. (line)'
@55 COV1    15.4 /
@25 'Cov. H.S. (tester)'
@55 COV2    15.4 /
@25 'Cov. H.S. (average)'
@55 COV3    15.4 /
@25 'Cov. F.S.'
@55 COV4    15.4 /
@25 's2A with F=0'
@55 S2A0    15.4 /
@25 's2A with F=1'
@55 S2A1    15.4 /
@25 's2D with F=0'
@55 S2D0    15.4 /
@25 's2D with F=1'
@55 S2D1    15.4 /
@25 'Contribution of lines'
@55 CL      15.2 @ 71 '%' /
@25 'Contribution of testers'
@55 CT      15.2 @ 71 '%' /
@25 'Contribution of (l x t)'
@55 CLXT    15.2 @ 71 '%' /
@24 R1;
RUN;

*********** HETEROSIS ANALYSIS ***********;
DATA MAINT; SET PMEAN; IF ID=2; RENAME VAR=LINE
PMEAN=FEMALE;
PROC SORT; BY LINE;
DATA REST; SET PMEAN; IF ID=1;  RENAME
VAR=TESTER PMEAN=MALE;
PROC SORT; BY TESTER;
PROC SORT DATA=HMEAN; BY LINE;
DATA ALL; MERGE HMEAN MAINT; BY LINE;
PROC SORT; BY TESTER;
DATA ALL; MERGE ALL REST; BY TESTER;
     MP=(FEMALE+MALE)/2;
     BP=MAX(FEMALE,MALE);
     DMP=HMEAN-MP;
     HMP=(DMP/MP)*100;
     DBP=HMEAN-BP;
     HBP=(DBP/BP)*100;

PROC TABULATE  FORMAT=7.2;
   CLASS LINE TESTER;
   VAR HMEAN FEMALE MALE MP BP DMP HMP DBP HBP;
   TITLE "HETEROSIS ANALYSIS FOR &&CHAR&K";
   TABLE LINE=' '*TESTER=' ', (HMEAN='Hybrid
       Mean (F1)' FEMALE='Female' MALE='Male'
       MP='Parent Mean (MP)'
       BP='Better Parent (BP)'
       DMP='Diff. from MP'
       HMP='Heterosis over MP (%)'
       DBP='Diff. from BP'
       HBP='Heterosis over BP (%)')*MEAN=' '/
       BOX='Line Tester' RTS=15;

%END;
RUN;
%MEND LINXTES;
```

APPENDIX B. PROGRAM OUTPUTS

A. ANALYSIS OF VARIANCE TABLE

ANALYSIS OF VARIANCE FOR GRAIN YIELD (KG/HA)

Source of Variation	df	Sum of Squares	Mean Square	F	Prob
REP	2	1455214.7442	727607.3721	5.10	0.0083
VAR	42	81725258.3566	1945839.4847	13.63	0.0000
C VS P VS H	2	25940241.6769	12970120.8385	90.83	0.0000
C VS (P,H)	1	3971791.7241	3971791.7241	27.82	0.0000
P VS H	1	21968449.9528	21968449.9528	153.85	0.0000
CHECKS (C)	3	2807278.0000	935759.3333	6.55	0.0008
PARENT (P)	10	9083068.0606	908306.8061	6.36	0.0000
HYBRID (H)	27	43894670.6190	1625728.5414	11.39	0.0000
LINE	6	20995238.2857	3499206.3810	3.07	0.0299
TESTER	3	2359246.2381	786415.4127	< 1	
LINE x TESTER	18	20540186.0952	1141121.4497	7.99	0.0000
ERROR	84	11994357.9225	142789.9753		

CV = 20.53%

B. TABLE OF PARENT MEANS AND THEIR RESPECTIVE GENERAL COMBINING ABILITY (PARTIAL)

TABLE OF PARENT MEANS AND GCA FOR GRAIN YIELD (KG/HA)
(AVE. OF 3 REPS)

Variety		Parent Mean	GCA
TESTER	1	796.333	228.2143
	2	370.000	52.8810
	3	767.667	-235.3095
	4	1168.667	-45.7857
LINE	5	1984.333	-635.3571
	6	1019.667	369.2262
	7	1799.000	760.8929
	8	1919.000	428.1429

C. TABLE OF HYBRID MEANS AND THEIR RESPECTIVE SPECIFIC COMBINING ABILITY (PARTIAL)

TABLE OF HYBRID MEANS AND SCA FOR GRAIN YIELD (KG/HA)
(AVE. OF 3 REPS)

Variety	Hybrid Mean	SCA
12	1762.667	1.7857
13	2516.000	-249.4643
14	3275.000	117.8690
15	2649.667	-174.7143
16	2620.000	497.0357
17	1562.667	-224.4643
18	2387.667	31.9524
19	1656.000	70.4524
20	2802.667	212.5357

D. STANDARD ERRORS OF COMBINING ABILITY EFFECTS

STANDARD ERRORS OF COMBINING ABILITY EFFECTS
FOR GRAIN YIELD (KG/HA)

Parameter	Estimate
S.E. (gca for line)	109.0833
S.E. (gca for tester)	82.4592
S.E. (sca for effects)	218.1666
S.E. (gi-gj) line	154.2671
S.E. (gi-gj) tester	116.6149
S.E. (sij-skl)	308.5341
Cov. H.S. (line)	196507.0776
Cov. H.S. (tester)	-16890.7637
Cov. H.S. (average)	10769.0465
Cov. F.S.	537426.4023
s2A with F=0	172304.7437
s2A with F=1	43076.1859
s2D with F=0	5324434.5305
s2D with F=1	1331108.6326
Contribution of lines	47.83 %
Contribution of testers	5.37 %
Contribution of (l x t)	46.79 %

E. HETEROSIS ANALYSIS (PARTIAL)

HETEROSIS ANALYSIS FOR GRAIN YIELD (KG/HA)

Line	Tester	Hybrid Mean (F1)	Female	Male	Parent Mean (MP)	Better Parent (BP)	Diff. from MP	Hetero-sis over MP (%)	Diff. from BP	Hetero-sis over BP (%)
5	1	1762.67	1984.33	796.33	1390.33	1984.33	372.33	26.78	-221.67	-11.17
	2	1656.00	1984.33	370.00	1177.17	1984.33	478.83	40.68	-328.33	-16.55
	3	1081.67	1984.33	767.67	1376.00	1984.33	-294.33	-21.39	-902.67	-45.49
	4	1630.33	1984.33	1168.67	1576.50	1984.33	53.83	3.41	-354.00	-17.84
6	1	2516.00	1019.67	796.33	908.00	1019.67	1608.00	177.09	1496.33	146.75
	2	2802.67	1019.67	370.00	694.83	1019.67	2107.83	303.36	1783.00	174.86
	3	2181.00	1019.67	767.67	893.67	1019.67	1287.33	144.05	1161.33	113.89
	4	2649.33	1019.67	1168.67	1094.17	1168.67	1555.17	142.13	1480.67	126.70
7	1	3275.00	1799.00	796.33	1297.67	1799.00	1977.33	152.38	1476.00	82.05
	2	2979.00	1799.00	370.00	1084.50	1799.00	1894.50	174.69	1180.00	65.59
	3	2367.67	1799.00	767.67	1283.33	1799.00	1084.33	84.49	568.67	31.61
	4	3094.00	1799.00	1168.67	1483.83	1799.00	1610.17	108.51	1295.00	71.98
8	1	2649.67	1919.00	796.33	1357.67	1919.00	1292.00	95.16	730.67	38.08
	2	3442.00	1919.00	370.00	1144.50	1919.00	2297.50	200.74	1523.00	79.36
	3	1460.00	1919.00	767.67	1343.33	1919.00	116.67	8.68	-459.00	-23.92
	4	2833.00	1919.00	1168.67	1543.83	1919.00	1289.17	83.50	914.00	47.63
9	1	2620.00	708.67	796.33	752.50	796.33	1867.50	248.17	1823.67	229.01
	2	1910.67	708.67	370.00	539.33	708.67	1371.33	254.26	1202.00	169.61
	3	887.33	708.67	767.67	738.17	767.67	149.17	20.21	119.67	15.59
	4	2161.00	708.67	1168.67	938.67	1168.67	1222.33	130.22	992.33	84.91
10	1	1562.67	1068.67	796.33	932.50	1068.67	630.17	67.58	494.00	46.23
	2	1217.00	1068.67	370.00	719.33	1068.67	497.67	69.18	148.33	13.88
	3	2439.33	1068.67	767.67	918.17	1068.67	1521.17	165.67	1370.67	128.26

CONTACT INFORMATION

Author Name: VIOLETA BARTOLOME
Company: International Rice Research Institute
Address: MCPO Box 3127
City state ZIP: Makati City 1271
Country: Philippines
Work Phone: (632) 845-0569 (x437)
Fax: (632) 891-1292
Email: v.bartolome@cgiar.org
Web: http://www.cgiar.org/irri